

# NAG Library Routine Document

## E04JYF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

E04JYF is an easy-to-use quasi-Newton algorithm for finding a minimum of a function  $F(x_1, x_2, \dots, x_n)$ , subject to fixed upper and lower bounds of the independent variables  $x_1, x_2, \dots, x_n$ , using function values only.

It is intended for functions which are continuous and which have continuous first and second derivatives (although it will usually work even if the derivatives have occasional discontinuities).

### 2 Specification

```

SUBROUTINE E04JYF (N, IBOUND, FUNCT1, BL, BU, X, F, IW, LIW, W, LW, IUSER,      &
                  RUSER, IFAIL)
INTEGER           N, IBOUND, IW(LIW), LIW, LW, IUSER(*), IFAIL
REAL (KIND=nag_wp) BL(N), BU(N), X(N), F, W(LW), RUSER(*)
EXTERNAL         FUNCT1

```

### 3 Description

E04JYF is applicable to problems of the form:

$$\text{Minimize } F(x_1, x_2, \dots, x_n) \quad \text{subject to} \quad l_j \leq x_j \leq u_j, \quad j = 1, 2, \dots, n$$

when derivatives of  $F(x)$  are unavailable.

Special provision is made for problems which actually have no bounds on the  $x_j$ , problems which have only non-negativity bounds and problems in which  $l_1 = l_2 = \dots = l_n$  and  $u_1 = u_2 = \dots = u_n$ . You must supply a subroutine to calculate the value of  $F(x)$  at any point  $x$ .

From a starting point you supplied there is generated, on the basis of estimates of the gradient and the curvature of  $F(x)$ , a sequence of feasible points which is intended to converge to a local minimum of the constrained function. An attempt is made to verify that the final point is a minimum.

A typical iteration starts at the current point  $x$  where  $n_z$  (say) variables are free from both their bounds. The projected gradient vector  $g_z$ , whose elements are finite difference approximations to the derivatives of  $F(x)$  with respect to the free variables, is known. A unit lower triangular matrix  $L$  and a diagonal matrix  $D$  (both of dimension  $n_z$ ), such that  $LDL^T$  is a positive definite approximation of the matrix of second derivatives with respect to the free variables (i.e., the projected Hessian) are also held. The equations

$$LDL^T p_z = -g_z$$

are solved to give a search direction  $p_z$ , which is expanded to an  $n$ -vector  $p$  by an insertion of appropriate zero elements. Then  $\alpha$  is found such that  $F(x + \alpha p)$  is approximately a minimum (subject to the fixed bounds) with respect to  $\alpha$ ;  $x$  is replaced by  $x + \alpha p$ , and the matrices  $L$  and  $D$  are updated so as to be consistent with the change produced in the estimated gradient by the step  $\alpha p$ . If any variable actually reaches a bound during the search along  $p$ , it is fixed and  $n_z$  is reduced for the next iteration. Most iterations calculate  $g_z$  using forward differences, but central differences are used when they seem necessary.

There are two sets of convergence criteria – a weaker and a stronger. Whenever the weaker criteria are satisfied, the Lagrange multipliers are estimated for all the active constraints. If any Lagrange multiplier estimate is significantly negative, then one of the variables associated with a negative Lagrange multiplier estimate is released from its bound and the next search direction is computed in the extended subspace (i.e.,  $n_z$  is increased). Otherwise minimization continues in the current subspace provided that this is

practicable. When it is not, or when the stronger convergence criteria are already satisfied, then, if one or more Lagrange multiplier estimates are close to zero, a slight perturbation is made in the values of the corresponding variables in turn until a lower function value is obtained. The normal algorithm is then resumed from the perturbed point.

If a saddle point is suspected, a local search is carried out with a view to moving away from the saddle point. A local search is also performed when a point is found which is thought to be a constrained minimum.

## 4 References

Gill P E and Murray W (1976) Minimization subject to bounds on the variables *NPL Report NAC 72*  
National Physical Laboratory

## 5 Parameters

1: N – INTEGER *Input*  
*On entry:* the number  $n$  of independent variables.  
*Constraint:*  $N \geq 1$ .

2: IBOUND – INTEGER *Input*  
*On entry:* indicates whether the facility for dealing with bounds of special forms is to be used.  
It must be set to one of the following values:

IBOUND = 0

If you are supplying all the  $l_j$  and  $u_j$  individually.

IBOUND = 1

If there are no bounds on any  $x_j$ .

IBOUND = 2

If all the bounds are of the form  $0 \leq x_j$ .

IBOUND = 3

If  $l_1 = l_2 = \dots = l_n$  and  $u_1 = u_2 = \dots = u_n$ .

3: FUNCT1 – SUBROUTINE, supplied by the user. *External Procedure*  
You must supply FUNCT1 to calculate the value of the function  $F(x)$  at any point  $x$ . It should be tested separately before being used with E04JYF (see the E04 Chapter Introduction).

The specification of FUNCT1 is:

```
SUBROUTINE FUNCT1 (N, XC, FC, IUSER, RUSER)
```

```
INTEGER          N, IUSER(*)
```

```
REAL (KIND=nag_wp) XC(N), FC, RUSER(*)
```

1:	N – INTEGER	<i>Input</i>
	<i>On entry:</i> the number $n$ of variables.	
2:	XC(N) – REAL (KIND=nag_wp) array	<i>Input</i>
	<i>On entry:</i> the point $x$ at which the function value is required.	
3:	FC – REAL (KIND=nag_wp)	<i>Output</i>
	<i>On exit:</i> the value of the function $F$ at the current point $x$ .	

4:	IUSER(*) – INTEGER array	<i>User Workspace</i>
5:	RUSER(*) – REAL (KIND=nag_wp) array	<i>User Workspace</i>
FUNCT1 is called with the parameters IUSER and RUSER as supplied to E04JYF. You are free to use the arrays IUSER and RUSER to supply information to FUNCT1 as an alternative to using COMMON global variables.		

FUNCT1 must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub)program from which E04JYF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

- 4: BL(N) – REAL (KIND=nag\_wp) array *Input/Output*  
*On entry:* the lower bounds  $l_j$ .  
 If IBOUND is set to 0, you must set BL( $j$ ) to  $l_j$ , for  $j = 1, 2, \dots, n$ . (If a lower bound is not specified for a particular  $x_j$ , the corresponding BL( $j$ ) should be set to  $-10^6$ .)  
 If IBOUND is set to 3, you must set BL(1) to  $l_1$ ; E04JYF will then set the remaining elements of BL equal to BL(1).  
*On exit:* the lower bounds actually used by E04JYF.
- 5: BU(N) – REAL (KIND=nag\_wp) array *Input/Output*  
*On entry:* the upper bounds  $u_j$ .  
 If IBOUND is set to 0, you must set BU( $j$ ) to  $u_j$ , for  $j = 1, 2, \dots, n$ . (If an upper bound is not specified for a particular  $x_j$ , the corresponding BU( $j$ ) should be set to  $10^6$ .)  
 If IBOUND is set to 3, you must set BU(1) to  $u_1$ ; E04JYF will then set the remaining elements of BU equal to BU(1).  
*On exit:* the upper bounds actually used by E04JYF.
- 6: X(N) – REAL (KIND=nag\_wp) array *Input/Output*  
*On entry:* X( $j$ ) must be set to an estimate of the  $j$ th component of the position of the minimum, for  $j = 1, 2, \dots, n$ .  
*On exit:* the lowest point found during the calculations. Thus, if IFAIL = 0 on exit, X( $j$ ) is the  $j$ th component of the position of the minimum.
- 7: F – REAL (KIND=nag\_wp) *Output*  
*On exit:* the value of  $F(x)$  corresponding to the final point stored in X.
- 8: IW(LIW) – INTEGER array *Output*  
*On exit:* if IFAIL = 0, 3 or 5, the first N elements of IW contain information about which variables are currently on their bounds and which are free. Specifically, if  $x_i$  is:  
 – fixed on its upper bound, IW( $i$ ) is  $-1$ ;  
 – fixed on its lower bound, IW( $i$ ) is  $-2$ ;  
 – effectively a constant (i.e.,  $l_j = u_j$ ), IW( $i$ ) is  $-3$ ;  
 – free, IW( $i$ ) gives its position in the sequence of free variables.  
 In addition, IW(N + 1) contains the number of free variables (i.e.,  $n_z$ ). The rest of the array is used as workspace.

- 9: LIW – INTEGER *Input*  
*On entry:* the dimension of the array IW as declared in the (sub)program from which E04JYF is called.  
*Constraint:*  $LIW \geq N + 2$ .
- 10: W(LW) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* if IFAIL = 0, 3 or 5,  $W(i)$  contains a finite difference approximation to the  $i$ th element of the projected gradient vector  $g_z$ , for  $i = 1, 2, \dots, N$ . In addition,  $W(N + 1)$  contains an estimate of the condition number of the projected Hessian matrix (i.e.,  $k$ ). The rest of the array is used as workspace.
- 11: LW – INTEGER *Input*  
*On entry:* the dimension of the array W as declared in the (sub)program from which E04JYF is called.  
*Constraint:*  $LW \geq \max(N \times (N - 1)/2 + 12 \times N, 13)$ .
- 12: IUSER(\*) – INTEGER array *User Workspace*  
 13: RUSER(\*) – REAL (KIND=nag\_wp) array *User Workspace*  
 IUSER and RUSER are not used by E04JYF, but are passed directly to FUNCT1 and may be used to pass information to this routine as an alternative to using COMMON global variables.
- 14: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if  $IFAIL \neq 0$  on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**  
*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

**Note:** E04JYF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL = 1

- On entry,  $N < 1$ ,
- or IBOUND < 0,
- or IBOUND > 3,
- or IBOUND = 0 and  $BL(j) > BU(j)$  for some  $j$ ,
- or IBOUND = 3 and  $BL(1) > BU(1)$ ,
- or  $LIW < N + 2$ ,
- or  $LW < \max(13, 12 \times N + N \times (N - 1)/2)$ .

IFAIL = 2

There have been  $400 \times n$  function evaluations, yet the algorithm does not seem to be converging. The calculations can be restarted from the final point held in X. The error may also indicate that  $F(x)$  has no minimum.

IFAIL = 3

The conditions for a minimum have not all been met but a lower point could not be found and the algorithm has failed.

IFAIL = 4

An overflow has occurred during the computation. This is an unlikely failure, but if it occurs you should restart at the latest point given in X.

IFAIL = 5

IFAIL = 6

IFAIL = 7

IFAIL = 8

There is some doubt about whether the point  $x$  found by E04JYF is a minimum. The degree of confidence in the result decreases as IFAIL increases. Thus, when IFAIL = 5 it is probable that the final  $x$  gives a good estimate of the position of a minimum, but when IFAIL = 8 it is very unlikely that the routine has found a minimum.

IFAIL = 9

In the search for a minimum, the modulus of one of the variables has become very large ( $\sim 10^6$ ). This indicates that there is a mistake in FUNCT1, that your problem has no finite solution, or that the problem needs rescaling (see Section 8).

IFAIL = 10

The computed set of forward-difference intervals (stored in  $W(9 \times N + 1), W(9 \times N + 2), \dots, W(10 \times N)$ ) is such that  $X(i) + W(9 \times N + i) \leq X(i)$  for some  $i$ .

This is an unlikely failure, but if it occurs you should attempt to select another starting point.

If you are dissatisfied with the result (e.g., because IFAIL = 5, 6, 7 or 8), it is worth restarting the calculations from a different starting point (not the point at which the failure occurred) in order to avoid the region which caused the failure. If persistent trouble occurs and the gradient can be calculated, it may be advisable to change to a routine which uses gradients (see the E04 Chapter Introduction).

## 7 Accuracy

A successful exit (IFAIL = 0) is made from E04JYF when (B1, B2 and B3) or B4 hold, and the local search confirms a minimum, where

$$B1 \equiv \alpha^{(k)} \times \|p^{(k)}\| < (x_{tol} + \sqrt{\epsilon}) \times \left(1.0 + \|x^{(k)}\|\right)$$

$$B2 \equiv \left|F^{(k)} - F^{(k-1)}\right| < (x_{tol}^2 + \epsilon) \times \left(1.0 + \left|F^{(k)}\right|\right)$$

$$B3 \equiv \|g_z^{(k)}\| < \left(\epsilon^{1/3} + x_{tol}\right) \times \left(1.0 + \left|F^{(k)}\right|\right)$$

$$B4 \equiv \|g_z^{(k)}\| < 0.01 \times \sqrt{\epsilon}.$$

(Quantities with superscript  $k$  are the values at the  $k$ th iteration of the quantities mentioned in Section 3,  $x_{tol} = 100\sqrt{\epsilon}$ ,  $\epsilon$  is the *machine precision* and  $\|\cdot\|$  denotes the Euclidean norm. The vector  $g_z$  is returned in the array W.)

If IFAIL = 0, then the vector in X on exit,  $x_{sol}$ , is almost certainly an estimate of the position of the minimum,  $x_{true}$ , to the accuracy specified by  $x_{tol}$ .

If IFAIL = 3 or 5,  $x_{sol}$  may still be a good estimate of  $x_{true}$ , but the following checks should be made. Let  $k$  denote an estimate of the condition number of the projected Hessian matrix at  $x_{sol}$ . (The value of  $k$  is returned in  $W(N + 1)$ ). If

- (i) the sequence  $\{F(x^{(k)})\}$  converges to  $F(x_{\text{sol}})$  at a superlinear or a fast linear rate,
- (ii)  $\|g_z(x_{\text{sol}})\|^2 < 10.0 \times \epsilon$ , and
- (iii)  $k < 1.0/\|g_z(x_{\text{sol}})\|$ ,

then it is almost certain that  $x_{\text{sol}}$  is a close approximation to the position of a minimum. When (ii) is true, then usually  $F(x_{\text{sol}})$  is a close approximation to  $F(x_{\text{true}})$ .

When a successful exit is made then, for a computer with a mantissa of  $t$  decimals, one would expect to get about  $t/2 - 1$  decimals accuracy in  $x$  and about  $t - 1$  decimals accuracy in  $F$ , provided the problem is reasonably well scaled.

## 8 Further Comments

The number of iterations required depends on the number of variables, the behaviour of  $F(x)$  and the distance of the starting point from the solution. The number of operations performed in an iteration of E04JYF is roughly proportional to  $n^2$ . In addition, each iteration makes at least  $m + 1$  calls of FUNCT1, where  $m$  is the number of variables not fixed on bounds. So, unless  $F(x)$  can be evaluated very quickly, the run time will be dominated by the time spent in FUNCT1.

Ideally the problem should be scaled so that at the solution the value of  $F(x)$  and the corresponding values of  $x_1, x_2, \dots, x_n$  are each in the range  $(-1, +1)$ , and so that at points a unit distance away from the solution,  $F$  is approximately a unit value greater than at the minimum. It is unlikely that you will be able to follow these recommendations very closely, but it is worth trying (by guesswork), as sensible scaling will reduce the difficulty of the minimization problem, so that E04JYF will take less computer time.

## 9 Example

To minimize

$$F = (x_1 + 10x_2)^2 + 5(x_3 - x_4)^2 + (x_2 - 2x_3)^4 + 10(x_1 - x_4)^4$$

subject to

$$\begin{aligned} 1 &\leq x_1 \leq 3 \\ -2 &\leq x_2 \leq 0 \\ 1 &\leq x_4 \leq 3, \end{aligned}$$

starting from the initial guess  $(3, -1, 0, 1)$ .

### 9.1 Program Text

```
! E04JYF Example Program Text
! Mark 24 Release. NAG Copyright 2012.
! Module e04jyfe_mod

! E04JYF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
! Use nag_library, Only: nag_wp
! .. Implicit None Statement ..
! Implicit None
! .. Parameters ..
! Integer, Parameter :: n = 4, nout = 6
! Integer, Parameter :: liw = n + 2
! Integer, Parameter :: lw = n*(n-1)/2 + 12*n
Contains
! Subroutine funct1(n,xc,fc,iuser,ruser)
! Routine to evaluate objective function.

! .. Scalar Arguments ..
! Real (Kind=nag_wp), Intent (Out) :: fc
! Integer, Intent (In) :: n
```

```

!      .. Array Arguments ..
      Real (Kind=nag_wp), Intent (Inout)  :: ruser(*)
      Real (Kind=nag_wp), Intent (In)     :: xc(n)
      Integer, Intent (Inout)            :: iuser(*)
!      .. Local Scalars ..
      Real (Kind=nag_wp)                 :: x1, x2, x3, x4
!      .. Executable Statements ..
      x1 = xc(1)
      x2 = xc(2)
      x3 = xc(3)
      x4 = xc(4)
      fc = (x1+10.0_nag_wp*x2)**2 + 5.0_nag_wp*(x3-x4)**2 + &
           (x2-2.0_nag_wp*x3)**4 + 10.0_nag_wp*(x1-x4)**4

      Return

      End Subroutine funct1
End Module e04jyfe_mod
Program e04jyfe

!      E04JYF Example Main Program

!      .. Use Statements ..
      Use nag_library, Only: e04jyf, nag_wp
      Use e04jyfe_mod, Only: funct1, liw, lw, n, nout
!      .. Implicit None Statement ..
      Implicit None
!      .. Local Scalars ..
      Real (Kind=nag_wp)                 :: f
      Integer                             :: ibound, ifail
!      .. Local Arrays ..
      Real (Kind=nag_wp)                 :: bl(n), bu(n), ruser(n), w(lw), &
                                         x(n)
      Integer                             :: iuser(n), iw(6)
!      .. Executable Statements ..
      Write (nout,*) 'E04JYF Example Program Results'
      Flush (nout)

      x(1:n) = (/3.0_nag_wp,-1.0_nag_wp,0.0_nag_wp,1.0_nag_wp/)
      ibound = 0

!      X(3) is unconstrained, so we set BL(3) to a large negative
!      number and BU(3) to a large positive number.

      bl(1:n) = (/1.0_nag_wp,-2.0_nag_wp,-1.0E6_nag_wp,1.0_nag_wp/)
      bu(1:n) = (/3.0_nag_wp,0.0_nag_wp,1.0E6_nag_wp,3.0_nag_wp/)

      ifail = -1
      Call e04jyf(n,ibound,funct1,bl,bu,x,f,iw,liw,w,lw,iuser,ruser,ifail)

      Select Case (ifail)
      Case (0,2:)
        Write (nout,*)
        Write (nout,99999) 'Function value on exit is ', f
        Write (nout,99998) 'at the point', x(1:n)
      End Select

99999 Format (1X,A,F8.4)
99998 Format (1X,A,4F9.4)
      End Program e04jyfe

```

## 9.2 Program Data

None.

### 9.3 Program Results

E04JYF Example Program Results

\*\* It is probable that a local minimum has been found,

\*\* but it cannot be guaranteed.

\*\* ABNORMAL EXIT from NAG Library routine E04JYF: IFAIL = 5

\*\* NAG soft failure - control returned

Function value on exit is 2.4338

at the point 1.0000 -0.0852 0.4093 1.0000

---